Here we “prove” (for $l = 0, ..., 4$) that, defining:
$\hat{q} = (\sin(\theta)\sin(\phi), \sin(\theta)\cos(\phi), \cos(\theta))$,
$\hat{k} = (\sin(\theta')\sin(\phi'), \sin(\theta')\cos(\phi'), \cos(\theta'))$,
one finds:
$(2l+1) P_l(\hat{k}\cdot\hat{q}) = 4\pi \sum_{M=-l}^{M=+l} Y_l^M(\theta, \phi)\, Y_l^M(\theta', \phi')^*$
This result is called the *Addition theorem for Spherical Harmonics.* A proof for arbitrary l can be found in J.D. Jackson, *Classical Electrodynamics*, 3ed., pp. 110-111. [Sec. 3.6].

In[148]:=
```
qunit = {Sin[θ] Sin[ϕ], Sin[θ] Cos[ϕ], Cos[θ]};
kunit = {Sin[θp] Sin[ϕp], Sin[θp] Cos[ϕp], Cos[θp]};
Table[Simplify[((2 l + 1) LegendreP[l, kunit.qunit]) /
    (4 π Sum[SphericalHarmonicY[l, M, θ, ϕ] Conjugate[SphericalHarmonicY[l, M, θp, ϕp]],
      {M, -l, l}]), Assumptions → {{θ, θp, ϕ, ϕp} ∈ Reals}], {l, 0, 4}]
```

Out[150]=
```
{1, 1, 1, 1, 1}
```

The other result that we showed yesterday was that:
$\sum_{M=-l}^{M=+l} Y_l^M(\theta', \phi')^* \int \hat{q}\, Y_l^M(\theta, \phi)\, d\Omega_{\hat{q}} = \delta_{l,1}\hat{k}$
Here we can check that this is true for $l = 1$ (the output is indeed $\hat{k}$) that for a few $l \neq 0$ it gives zero. But the proof can be done directly by hand.

In[154]:=
```
Table[FullSimplify[Sum[SphericalHarmonicY[l, M, θp, ϕp] Integrate[
      Sin[θ] Conjugate[SphericalHarmonicY[l, M, θ, ϕ]] qunit, {ϕ, 0, 2 π}, {θ, 0, π}],
    {M, -l, l}], Assumptions → {{θp, ϕp} ∈ Reals}], {l, 0, 3}]
```

Out[154]=
```
{{0, 0, 0}, {Sin[θp] Sin[ϕp], Cos[ϕp] Sin[θp], Cos[θp]}, {0, 0, 0}, {0, 0, 0}}
```

In[159]:=

$$\texttt{qunitY} = \left(\frac{1}{2}\sqrt{\frac{3}{\pi}}\right)^{-1} \texttt{FullSimplify}\Big[\Big\{-\frac{\texttt{SphericalHarmonicY[1, 1, θ, ϕ] + SphericalHarmonicY[1, -1, θ, ϕ]}}{\sqrt{2}\, i},$$
$$-\frac{\texttt{SphericalHarmonicY[1, 1, θ, ϕ] - SphericalHarmonicY[1, -1, θ, ϕ]}}{\sqrt{2}},$$
$$\texttt{SphericalHarmonicY[1, 0, θ, ϕ]}\Big\}\Big]$$

Out[159]=
```
{Sin[θ] Sin[ϕ], Cos[ϕ] Sin[θ], Cos[θ]}
```

In[161]:=
```
SphericalHarmonicY[1, +1, θ, ϕ]
SphericalHarmonicY[1, -1, θ, ϕ]
SphericalHarmonicY[1, 0, θ, ϕ]
```

Out[161]= $-\frac{1}{2} e^{i \phi} \sqrt{\frac{3}{2 \pi}} \text{Sin}[\theta]$

Out[162]= $\frac{1}{2} e^{-i \phi} \sqrt{\frac{3}{2 \pi}} \text{Sin}[\theta]$

Out[163]= $\frac{1}{2} \sqrt{\frac{3}{\pi}} \text{Cos}[\theta]$